Vol. I No. 5

Single Copies $2

the RAINBOW

5803 Timber Ridge Drive • Prospect, KY 40059

The Monthly Magazine for Color Computer Users

TANDY DISC SYSTEM IS NOW AVAILABLE

You read it first here. (Maybe we should gloat and add "of course".) It was in July.

Commercials for the RAINBOW aside, the Tandy Disc system is a reality. We know, because there is one sitting right here next to our Color Computer. (And we didn't pull any strings to get it, either.) By now, you can probably see one at your local Radio Shack computer center or store.

As we wrote earlier, the Disc Operating System is based in ROM -- a ROM Cart. It plugs into the CART port of the Color Computer. Drive "0" is the first step for a system. There are three more drives available. Our earlier information on prices was accurate -- but we were $4 low on drive 0. Drives 1, 2 and 3 were on target. But you can get that information in the new catalog.

For your $599, you get a drive, a drive controller (which is, actually, a large-size ROM Cart), a ribbon cable that hooks the cart up with drives 0 and 1, and manual and a free (!) disk. Incidentally, the box all this comes in says "custom mfg. in Japan."

The whole thing is packaged very well and comes in two boxes inside the shipping box. Obviously, Tandy will use the same packaging materials for the other drives. The manual is in the familiar multi-colored format of "Getting Started" and "Going Ahead." Its called "Color Computer Disc System" (Should it be "Spinning Along..."?)

(Continued on Page 2)

A REAL BIG ONE!

There are a lot of goodies in the November issue of the RAINBOW! We hope you enjoy them.

There is a Videotex downloading routine (that can be used with any host computer system), our basketball stats program, a takeout on Tandy's new disc drive, a biorhythm program ... and a whole lot more.

Next month ... some special seasonal stuff is comtemplated. But, too, there's a great program that will allow you to make backup copies of your ROM packs, the beginning of a series on role-playing game aids, reviews and more tutorials, helps, hints and tips.

We're really gratified by the response to the RAINBOW. Our subscription list is really growing -- and we like to think that's because you like what you see. Thanks for your support!

We'd also appreciate your help in spreading the word about the RAINBOW. And, although we know it sounds trite, please patronize our advertisers. After all, they care enough about YOU and YOUR computer to advertise in the ONLY monthly magazine devoted exclusively to the Color Computer.

DISC (Continued from pg. 1)

and features a new character to go with the drawings of the Color Computer -- a disc with arms and legs. As usual, the documentation is excellent. It is sort of a cross between the simple (in the early going) to a little complex (when sequential and direct access formats are discussed). We do wish there were a little more detail -- sans embedded formats -- near the end of the book.

As to the drive itself, it has worked flawlessly since its arrival several days ago. And, because the operating system is ROM-based, its use is transparent to the user. In other words, its there when you want it but it doesn't interfere with you in any way. And, because your cassette port isn't involved in this new system, you can CLOAD a program from tape and then SAVE it onto disc directly. I had two substantial tapes full of games -- and I transferred them one at a time with total ease. Now they are on disc, and totally accessable in just seconds. The longest program I have -- which took two and one-half minutes to load from tape -- loaded from disc in 12 seconds.

We were disappointed there is no AUTO command to generate automatic line numbers in the disc utilities. (If this bugs you as much as it does us, check the review on MASTER CONTROL in this issue for a solution.) Yet, one of the other big problems from the tape system has been solved. Through its VERIFY command, the disc will check itself to be sure your program was saved accurately. No more triple saves and prayers.

The DOS, which is referred to in the documentation as "Disc Basic," is pretty much the same as that available for the Model III, without the utilities. The discs are the same as those used by Model III.

PROGRAM QUICKIE...

WHERE AM I??

You probably know you can make multiple saves to tape simply by enclosing the CSAVE command in a loop. And, its a good idea to do this because there is no way to verify a tape write and if there is a tape problem, your copy can be lost.

Sometimes, though, those saves are long and you just sit there wondering "where am I?"

Here's a way to figure it out: Just put a counter in your direct command. Then, you'll know what CSAVE you are on. The direct command line looks like this: FOR X=1 TO 3:PRINT "ON SAVE"X:CSAVE "filename":NEXT X

If you want to save more or less than three programs, change the three in the X-loop. If you want to put space between the individual saves, add the following after the last colon above: MOTOR ON:FOR Y=1 TO 750 :NEXT Y:MOTOR OFF:

And yes, Virginia, it was written by Microsoft.

A disc system won't be for everyone. However, there are many applications for which a disc is the only way to go effectively and this system seems to have the features necessary for some pretty sophisticated programming. There are some other disc systems which will or are to become available, and we will keep you up to date on them as is possible.

BASKETBALL STATSKEEPER WILL HAVE THEM BOUNCING ALONG

When you gather your friends together for a TV game or two this coming basketball season -- set up a second TV and keep score for them with STATSKEEPER.

STATSKEEPER, in a larger version, was used last season to keep a full set of statistics on some games for the University of Louisville Cardinals. Because all statistics are kept up to date as of the time things occur, the final stats can be done instantly when the game ends. That program keeps track of every possible activity -- including minutes played.

STATSKEEPER is designed to be used at home (unless you want to take your Color Computer to the game). Because of that, and in an effort to get a professional-look box score display, the number of statistics categories are limited to field goals attempted and made and free throws attempted and made. STATSKEEPER keeps up with total points for each player and team and computes free throw and field goal percentages.

You use the "enter" option when a player comes into the game. This controls who is displayed in the box score. Be sure to "enter" the starting five. If you forget, the statistics will still accumulate -- they just won't be displayed.

It has been our experience that other statistical categories(such as rebounds, minutes played and so on) are most difficult to determine watching on TV. The field goal attempts might be a little difficult, but if you can concentrate on those, your statistics should come out all right.

(Continued on Page 8)

SOFTWARE REVIEW...

MASTER CONTROL

MASTER CONTROL is the best thing that's happened to tired fingers since molded keyboards.

One of the things that has bugged us about the Extended Color Basic command set is that there is no AUTO function included for ease of line numbering, especially when doing long lists of DATA statements. MASTER CONTROL gives us a line numbering function that works with ease -- and that's just one of some half a hundred things it does to make life easier for anyone who is writing programs.

This little gem (available from SOFT SECTOR MARKETING, 6250 Middlebelt, Garden City, MI 48135 for $24.95) automatically enters 50 commands at the touch of a finger. All you have to do is press the down arrow and any of the keys, which are marked with BASIC statements, and the statement you want appears on the screen. For example, if you have the automatic numbering in effect, you can write a line number and the DATA statement by pressing the "ENTER", down arrow and "D" keys only.

For those of us who frequently type "IF" as "OF" and who seem to always hit the "#" instead of the "$" in CHR$, this is a real plus that kills those syntax errors. The commands which need them (such as MID$) that have opening parentheses in them have the parentheses opening as part of the assigned statement.

In addition, the MOTOR, TRACE, AUDIO and RUN commands execute directly -- without your having to press the ENTER key. This really saves a lot of time.

Finally, there is what SOFT SECTOR describes as a "custom key", which you can program to be anything you want. This is a major bonus! If, for instance, you are running though a long series of IF statements which a couple of parameters, you can program the custom key for that whole list of parameters -- and save literally hours of typing. This is an excellent feature.

How do you keep track of what all the keys stand for? SOFT SECTOR supplies a template to overlay on the keyboard. That also means you can take it off if you want. While the template itself has a sticky backing, you can stick it on some cardboard and just cut holes for the keys. We did this and it worked very well.

MASTER CONTROL is written in machine code, so it is very fast. It only takes up 1100 bytes and can be relocated -- especially important if you have or anticipate upgrading to 32K. It does require 16K to operate, but does not require Extended

However a number of the commands are Extended commands. They, obviously, won't work without Extended.

MASTER CONTROL is a workmanlike program that does everything well. If you are doing any amount of your own programming, it is well worth the moderate cost in terms of typing hours saved alone.

MAILING NOTICE

From time to time, the RAINBOW may make its mailing list available to highly qualified firms which produce products for the Color Computer.

While we believe this information will be helpful to you, we respect your right to privacy and will not disclose your name or address to anyone if you will merely inform us that this is your desire.

Thank you.

SERIAL/PARALLEL CONVERTER

O.K., you're just not satisfied with either of the printers Radio Shack has out for the Color Computer -- either the Line Printer VII or the newest one, the Line Printer VIII.

Or, maybe, you just want to be able to switch printers at some time in the future and keep your present LP VII or VIII for now.

The choice is yours, but you can gain some real flexibility with the PI80C SERIAL/PARALLEL CONVERTER from THE MICRO WORKS (P.O. Box 1110, Del Mar, CA, 92014).

For $69.95, the PI80C will allow you to use the serial port of your Color Computer for your printer, but will convert the signals for use on the many parallel printers now on the market. That, by the way, includes virtually every major printer manufacturer. You can purchase conversion kits for a number of printers, but the cost is much more than is this little "black box" (which really IS a black box

We've had the PI80C up and running with an Epson printer for two months now. It has worked perfectly. Not a single hitch.

The PI80C comes with a four pin DIN plug which fits into the Color Computer's Serial port and a power cord which plugs into a standard socket. The other end is an edge card which has the "standard" Centronics configuration. You plug one end of a printer ribbon cable into the PI80C and the other into your printer. That's all!

Since the LP VII and LP VIII both have parallel connections, you can use your PI80C with one of those printers ... as well as with any other printer Radio Shack sells. You can also plug into an Epson, Centronics, Okidata, etc., etc.

TIMES SQUARE HEADER

So, you've written a neat program but you don't really have a good way to introduce it ... and you're tired, anyway.

Here's a quick way to use some of the more interesting capabilities of the Color Computer to produce a "header" for a program that's, as they say, fast and dirty.

This little gem will give you a Times Square-like message center on your screen. The words will just scroll along. And, from the code below, you'll see that it's easy to produce. You might even want to use it to leave messages for other members of the family!

For those without Extended Color Basic, change the STRING$ (16,32) in line 3 to 16 blank spaces. Line 11 can be deleted, its just there to keep the display neat in this example.

Here's the RPQ (RAINBOW Program Quickie) --with thanks to Joe Bennett--:

```
                                        ).
1 CLS0
3 A$="------- THIS PROGRAM IS
   BROUGHT TO YOU THROUGH THE
   RAINBOW...THE MONTHLY MAGAZ
   INE FOR COLOR COMPUTER USER
   S. YOU CAN PUT YOUR OWN MES
   SAGE HERE. -------"+STRING$
   (16,32)
5 FOR A=1 TO LEN(A$)-15:E1=E1
   +1:IF E1>4 THEN E1=1:E=1125
   :EE=1140
7 PRINT@328,MID$(A$,A,15);
9 SOUND RND(240),1:NEXT A
11 GOTO 11
```

In short, this fine piece of hardware gives you complete flexibility to buy whatever printer you wish without costing an arm and a leg for spec features.

This is a good buy.

HOW'S YOUR DAY?
NEED BIORHYTHM HELP?

Here's another program from JARB Software that will help you out in dealing with everyday life.

Its called JARBIORHYTHM and it will generate a complete chart for you. The chart can be as short as a day, or as long as a year. And, the chart will graphically display just how your biorhythms fluctuate over their different cycles.

You do need a printer for this program. But it comes out with a handsome display. We've heard that programs such as this can be a big hit at flea markets!

For those of you with tired fingers, JARBIORHYTHM can be obtained on a tape with PSYCHIC APTITUDE TEST (from RAINBOW Vol. I, No. 3) for $14.95 by writing JARB at 1169 Florida St., Imperial Beach, CA 92032. Please include $1 for postage and handling.

The Listing:

```
10 CLS:PRINT@196,TAB(11)"BIORHYT
HM":PRINTTAB(15)"BY":PRINTTAB(9)
"JARB SOFTWARE":PRINT:PRINT:PRIN
TTAB(4)"(C) JARB SOFTWARE 1981":
FORI=1TO2500:NEXTI
20 CLS:PRINT@160,"THIS PROGRAM I
S DESIGNED TO WORK";:PRINT"WITH
THE LINE PRINTER VII.":PRINT"WHI
LE IT WILL WORK WITH OTHER":PRIN
T"PRINTERS, IT MAY NOT FORMAT ON
":PRINT"THE PRINTER AS DESIGNED.
"
30 PRINT"YOU MUST HAVE A PRINTER
 ON-LINE":PRINT"TO OPERATE THIS
PROGRAM!!!!!!!!"
40 PRINT:LINEINPUT"TO BEGIN, PRE
SS <ENTER> KEY.";RD$
50 ' * * * * * * * * * * * * * *
60 ' *      JARBBIORHYTM      *
80 ' *           BY           *
90 ' *      JARB SOFTWARE     *
100 '*       WRITTEN  BY      *
110 '*      J. E. BENNETT     *
113 '*           AND          *
115 '*      JOHN L. URBAN     *
120 '*    (C) JARB SOFTWARE   *
130 '*          1981          *
140 '* * * * * * * * * * * * * *
150 PS=PEEK(65314):IFPS=40RPS=6T
HEN200ELSE160
160 CLS:PRINT@224,"PRINTER NOT O
N-LINE.....":PRINT"DO YOU
HAVE A PRINTER";:INPUTX$
170 IFLEFT$(X$,1)<>"Y"THEN200
180 CLS:PRINT@224,"YES?":PRINT"P
LEASE PLACE PRINTER ON-LINE.":LI
NEINPUT"PRESS ENTER WHEN ON-LINE
.";X$
190 PS=PEEK(65314):IFPS=40RPS=6T
HEN200ELSE160
200 CLEAR200
210 CLS:PRINT@224,"PRINTING TITL
E....PLEASE WAIT."
220 L=0:T=25:P=3.1415926535
230 PRINT#-2,CHR$(31):PRINT#-2,T
AB(16)"BIORHYTHM":PRINT#-2,TAB(1
9)"BY":PRINT#-2,TAB(14)"JARB SOF
TWARE"
240 PRINT#-2,CHR$(30);TAB(32)"(C
) JARB SOFTWARE 1981";CHR$(10);C
HR$(10);CHR$(10)
250 B$="YES":CLS:PRINT@224,"I AM
 NOW READY FOR FURTHER INPUT";:F
ORI=1TO1000:NEXTI
257 L=0:T=25:P=3.1415926535
260 CLS:PRINT@224," ":LINEINPUT"
WHAT IS YOUR NAME?";N$
270 CLS:PRINT@224,"":INPUT"THIS
CHART IS FOR HOW MANY DAYS";E1
275 GOSUB 1650
280 ZZ=0
300 CLS:PRINT@224,"ENTER BIRTH D
ATE.............."
310 GOSUB 550
313 GOSUB 1700
320 GOSUB720
330 JB=JD
340 CLS:PRINT@224,"ENTER START D
ATE FOR CHART....."
350 ZZ=1
360 GOSUB550
363 GOSUB 1750
370 GOSUB720
380 JC=JD
390 IF JC>=JB THEN 410
```

(Continued on Page 14)

COLOR METEOROIDS

The second program in SPECTRAL ASSOCIATES' Space Trilogy is called COLOR METEOROIDS, and it is -- like its brothers COLOR SPACE INVADERS and SPACE WAR -- a fast-action, machine language game that will bring hours of delight to devotees.

Although you need joysticks to play, you do not need Extended Color Basic. This can be a real plus for some, and it also shows just what you can do with machine language.

The game is like the arcade version, with a number of enhancements and plenty of action. Basically, you are the pilot of a ship navigating though the middle of a meteor swarm. You have to fire at the meteors, breaking them apart. They blast apart, but, then, you have to hit them again (and again!) to make them finally go away. Since there are 16 different levels of difficulty, this is a game in which the whole family can compete.

One of the things you can do with COLOR METEOROIDS that you can't do in the arcades is control the position of your gun. Here, you have full control -- as well as control of the ship itself. And for the really adept, you get a bonus ship for every 10,000 points. There's also a demonstration mode for cocktail party conversation.

Finally, there is an option that will allow your ship to move into hyperspace. Here, you just blip out of one location and into another. But, you must excercise caution -- sometimes various meanies in hyperspace can destroy your ship.

The graphics of COLOR METEOROIDS are excellent and the action is fast-paced. The sound effects, despite a disclaimer from SPECTRAL, are good, as well.

That, combined with special added effects such as difficulty levels, hyperspace, and an alien space ship, make COLOR METEOROIDS ($21.95 from SPECTRAL, 141 Harvard Ave., Tacoma, WA 98466) an excellent buy.

LETTERS TO RAINBOW...

Editor:

We enjoyed the RAINBOW very much. I'm into ways the Color Computer can help around the house and my husband is really into games.

Do you know of a PACMAN game that is available?

KATHY KLEINMAN
Overland Park, Kansas

* * * * *

Editor:

I am a first time computer user and feel frustrated by the magazines as all the information is written for other computers.

What I would really like is a list of functions for Model I and III and the same set of commands for my color computer.

JAMES RAUMER
Spokane, WA

Editor:

I recently purchased a copy of the RAINBOW ($3) and was quite pleased with the articles and especially with LASER STAR.

Just one comment about the last issue: In line 10 of the Graphics Printer, shouldn't '127' be '31'? Also, your choice of X for the vertical and Y for the horizonal coordinates is confusing. Normally, X is the horizonal.

KATHY GOETEL
Detroit

(Ed. Note: '31' works, too. Thanks for the information. As to the X and Y, you're right. It must have been a late night when that one was written!)

STATS (Continued from pg. 3)

Don't worry about length of names in the box score, as they will be chopped off to fit. We have made provision for games in which more than 100 points are scored, however, to satisfy you NBA fans.

All pregame entries (such as players names and numbers) are done through regular input statements. However, in order to speed execution of the program, all statistics are entered via INKEY$. The exception to this is the "Continue?" input when a box is being displayed. This allows you to retain the box on the screen for study while things are slow.

The only restriction in the program is that you can use only 12 players. This is simply in an effort to retain all players on the screen -- and we have left "room" in the DIM statements for 16 if you want more people and less heading information. It you opt for the 12 players, simply use 11 and label the 12th "others."

Two hints: If you hit the wrong letter or number while in the statistics collecting part of the program, just keep hitting the RETURN key until you get back to the "ENTER PLAYER'S TEAM" prompt. Once you have initialized the program, you can get back to that prompt by BREAKing and entering GOTO 210.

This program is written for Extended Color Basic, but the few things it uses from that instruction set are simple to convert. The STRING$(32,"=") is simply a line of 32 equals marks. LINE INPUT can be replaced by INPUT, but we used it for display purposes.

The listing:

```
 9 ' ***************************
10 '***************************
11 '**  STATSKEEPER PROGRAM **
12 '**                       **
13 '**        (c) 1981       **
14 '**                       **
15 '**     F A L S O F T     **
16 '***************************
17 '***************************
25 DIM HN$(16),HN(16),VN$(16),VN
(16),HG(16),HS(16),HF(16),HL(16)
,MP(16),HT(16),VT(16),VG(16),VS(
16),VF(16),VL(16),VE(16)
30 BS$="%          %  ## -##  ##
-##  ##"
31 BZ$=" PLAYER       FGM-FGA FTM
-FTA PTS"
32 BT$=" TOTALS       ## -##  ##
-## ###"
33 PC$="               ###.##% ##
#.##%"
35 CLS:PRINT:LINEINPUT "NAME THE
 HOME TEAM ";H$:LINEINPUT "NAME
THE VISITING TEAM ";V$
40 CLS:PRINTTAB(8)"--- HOME TEAM
 ---"
41 PRINT
45 INPUT "HOW MANY PLAYERS ARE E
LIGIBLE    TO PARTICIPATE FOR TH
E HOME     TEAM";SS
50 FOR X=1 TO SS
55 IL$=STRING$(32,"="):PRINTIL$;
60 PRINTH$" PLAYER'S NAME":PRINT
TAB(4);:INPUTHN$(X)
70 PRINTHN$(X)"'S NUMBER";:INPUT
HN(X)
80 NEXT X
90 CLS:PRINTTAB(6)"--- VISITING
TEAM ---"
95 INPUT "HOW MANY PLAYERS ARE E
LIGIBLE TO    PLAY FOR THE VISIT
ORS";VV
100 FOR X=1 TO VV
103 PRINTTAB(10)" (E)NTERED GAME
"
105 PRINTIL$;
110 PRINTV$" PLAYER'S NAME":PRIN
TTAB(4);:INPUTVN$(X)
120 PRINTVN$(X)"'S NUMBER";:INPU
TVN(X)
130 NEXT X
140 CLS:PRINT@163,"STATSKEEPER I
NITIALIZATION":PRINT"
COMPLETE"
150 PRINT@389,"PRESS <ENTER> TO
BEGIN";:PRINT"                SCO
REKEEPING";:PRINT@462," ";:LINEI
NPUTCN$:CLS
210 PRINT"---> ENTER PLAYER'S TE
AM <---":PRINT"     (H)OME OR (V
)ISITOR"
211 T$=INKEY$:IF T$="" THEN 211
212 CLS:IF T$="V" THEN 400ELSE P
RINTH$" STATISTIC"
```

DOWNLOADING FROM VIDEOTEX

By Jorge Mir

VIDEOAID is an enhancement to the VIDPRINT program which was published in last month's RAINBOW. It clears up some bugs, but it also allows you to download programs from CompuServe and BBS systems.

This program revises Radio Shack's VIDEOTEX. You need the VIDEOTEX program to make it work. But, by following the instructions, you will be able to view pages stored in memory, select any one page for viewing, copy any or all pages to tape, load pages previously stored to tape, obtain hard copy and download programs (and save them to tape).

You must first modify VIDEOTEX. If you have 16K, just load VIDEOTEX, type POKE2103,255 and <ENTER>. Then save the revision to tape by using CSAVEM "VIDEOTEX",1728,3839,1728. Use the new tape from now on.

Those with 32K can do a little more. First load VIDEOTEX. Then type PCLEAR1, load the VIDFIX program below, and run it. Then reload the corrected VIDEOTEX (the one generated to tape by VIDFIX).

Both systems can now run the revised VIDEOTEX. After you disconnect, RESET the computer to return to BASIC.

Now, load VIDEOAID (below). First type POKE25,6:POKE26,1 <ENTER>. Then load VIDEOAID with the normal CLOAD.

You can now type single letters to do the following: <P> Go back to previous page; <SPACE> Next page; <S> Save page being viewed to tape; <L> load previously saved page; <C> Change to specific page; <P> Print page being viewed; <D> Download. Make sure the printer is on before using <P>, the recorder before using <L> or <S>.

When you download, a white cursor will appear at the top left-hand corner of the screen. Use the arrow keys to move. If you move the cursor off the screen down, the next page will come up, and vise-versa.

Mark the program you wish to download by:

Press <B> for beginning. Move the cursor to the top of the first character and press the "B". You won't notice any change in the screen, however.

Mark the end of each line with an <L>. Place the cursor on top of the last character. Press "L".

Mark the end of the program. Use an <E> instead of an <L> as in the previous step. The program will then get ready to load onto tape. You will be prompted to assign a program name, and there will be a delay while the program reformats information for the tape recorder.

You can then RUN the program again or, if you wish to make a second copy, simply GOTO10, advance to the last page, type <D>, mark the last character with the <E> and it will be done again.

(Continued on Page 16)

GET AND PUT
High-Res Graphic Movement

By J. E. Pennett

How would you like to be able to move an image around the screen in BASIC. Sure, there are several ways to accomplish this. One way is to draw a picture, erase it, and then draw it in a new location.

The problem with this method is that it is very slow and creates a flickering effect. Tandy came to our rescue with the Color Computer graphic commands GET and PUT. These commands help you avoid the problems of speed and flicker, and still use just BASIC to write your programs. The manual for Extended Color Basic
but doen't explain their use to the average programmer. This article will explain how you can use these commands in your own programs.

First, there are sevral other things that must be done before you can call up GET and PUT. In order to illustrate these things, we will write a program to help us.

The first item on our list is to define the size array we will need for our picture. This array will be used each time we either GET a picture or PUT a picture. Its size is limited to how much memory we wish to dedicate to the picture. On a 16K RAM Color Computer, you have about 1400 elements of array storage. This is because each element uses 5 bytes of memory. At 1400 elements, you will use 7000 bytes (1400 elements * 5 bytes = 7K). In the highest mode of resolution, this leaves little for program use. To figure out the size array needed, just draw your picture, to scale, on paper. Then count how many pixels across it is (width) and how many pixels high it is (length).

As an example, enter this program and run it:

```
10 PMODE 4,1:PCLS:SCREEN 1,1
20 CIRCLE (128,96),5
30 GOTO 30
```

This program displays a circle in the center of the screen, 10 pixels across. The array size needed to cover this circle is 10 wide by 10 long. This gives us an array of 100 elements in size, and uses 500 bytes of memory. Using this size, add the following line to the example program you've typed in:

```
5 DIM A(9,9)
```

The GET command has a format that you must follow to insure accurate coverage of the picture you wish to store. This format is: GET (STARTPOINT) - (ENDPOINT), DESTINATION, G

The STARTPOINT is the upper left corner coordinates of your display. The ENDPOINT is the lower right corner coordinates of your display. The DESTINATION refers to the array name you have DIMensioned in Line 5, in this case, array A. The "G" will tell the computer that you wish to have the array stored with full graphics detail. This is optional, but we'll use it here and you should, too, for best results.

Now back to the program. Make the following changes:

```
20 A=128:B=96:CIRCLE (A,B),5
```

(Continued on Page 11)

What we've done is change the center point of the circle to be stated in terms of variables instead of fixed numeric points. Our circle will still be in the center of the screen, but we can now use the variables later without redefining their value.

Now, add the following line:

```
25 GET (A-5,B-5)-(A+5,B+5),A,G
```

As you can see, we have used the variables to allow us to set our start and end points more easily. The numeric equivalents of the variables would show we are now defining a square that is 10 pixels on a side. This will allow us to GET the complete circle we have previously drawn.

When the program is run, the circle will be stored in array A. Now that it is stored, what do we do with it? Well, how about drawing it in another location to prove it is stored in memory? We can do this by using the PUT command to place our circle anywhere we want on the screen.

The PUT command has the format: PUT (STARTPOINT) - (ENDPOINT), SOURCE, ACTION

The STARTPOINT and ENDPOINT are the new locations of the corners of our array. These can be any values that are within screen limits AND that hold the array to the size that was defined earlier in the program. In our example, the size is 10x10. The SOURCE is the name of the array we have stored earlier, array A. The ACTION is one of five options -- PSET, PRESET, AND, NOT and OR.

PSET is the action command we will use. This will allow all points (pixels) that are set in the original array to be set in the target location.

PRESET will reset any pixels in the target location that were originally set.

AND will compare the points in the original array with the points in the target location. If both are set, the new point will remain set. If one or the other is not set, the point will be reset.

NOT will reverse the state of every point in the target location. That is, if the point is set, it will be reset; if it is reset, it will be set.

OR will compare the points in the array with the points in the target location and if either is set, the screen will be set.

With all this in mind, add the following lines to your program:

```
30 A=A+INT(JOYSTK(0)/6.3)-5
   :B=B+INT(JOYSTK(1)/6.3)-5
40 PUT (A-5,B-5)-(A+5,B+5),A
   PSET
50 GOTO 30
```

Line 30 will now allow the right joystick to control movement in any direction at a maximum speed of up to five pixels at a time. This movement can be stopped by bringing the joystick to the exact center. The farther you move from the center, the faster the circle will move. Line 40 will PUT the circle at the location you have defined with the joystick. Line 50 will branch the program back to Line 30 for an update of the joystick location.

(Continued on Page 12)

GET (Continued from pg. 11)

Run the program and you will see that there are several problems with the operation. First, if you allow the circle to get too close to the screen edge, you will get a function call error. Let's take care of this problem by limiting the screen movement of the circle. We do this by adding:

```
31 IF A<=5 THEN A=5
32 IF A>=250 THEN A=250
33 IF B<=5 THEN B=5
34 IF B>=186 THEN B=186
```

After adding these lines, we can move the circle anywhere on the screen and it will stop or move along the edge if it gets close. The other problem we have is that the circle leaves a trail as it moves. While this can be used to create some pretty patterns, that isn't what we want to do.

We can eliminate the problem in two ways. Either we must enlarge the array to allow for the five pixel maximum movement of the circle, or we must decrease the size of the drawing to allow for a blank border around the drawing.

Both methods have drawbacks, but either can be used. If you increase the array size, you will use more memory and chance running into the nefarious OM error if your program is long. If you decrease the drawing size, you lose resolution, possibly affecting the effect you wanted. Since we have plenty of memory in this small program, we will use the first method. Make the following changes:

Change Line 5 to:

```
5 DIMA(20,20)
```

Change Line 25 to:

```
25 GET (A-10,B-10)-(A+10,B+10),
   A,G
```

Change Lines 31-34 to:

```
31 IF A<=10 THEN A=10
32 IF A>=245 THEN A=245
33 IF B<=10 THEN B=10
34 IF B>=181
```

Change Line 40 to:

```
40 PUT (A-10,B-10)-(A+10,B+10),
   A,PSET
```

Now we have a circle that will move under direction of the joystick and does not flicker. The speed of movement can be altered by changing the values in line 30, but you must also change the size of your array if you wish a faster movement. Otherwise, you will again leave a trail of picture parts on the screen.

Experiment with the other action command options. You might be pleasantly surprised with them.

Now that you have a better understanding of GET and PUT, look back through the last several issues of the RAINBOW and see how I used them in LASER STAR and HELO BATTLE. Remember, you have to use the same PMODE to PUT that you used to GET, or you may not obtain the results you tried to achieve.

Have fun with these and be sure to let me know via the RAINBOW how you are doing or if you have any questions or problems with which I can help.

---

the RAINBOW
PUBLISHED BY FALSOFT
5803 Timber Ridge Drive
Prospect, KY 40059

Lawrence C. Falk -- Editor

Subscriptions to the RAINBOW are $12 per year.

CORRECTION

On page 6 of RAINBOW Vol. I, No. 4, in the continuation of Al Morgan's VIDEO PRINT program, under the instructions, substitute the direct command POKE 25,6 for the PCLEAR1.

Al says this will make the program work correctly.

6809 Color Computer

EDITOR, ASSEMBLER AND MORE!

TIRED OF WAITING FOR SOFTWARE DELIVERY?

WE SHIP FROM STOCK!

NEW!

SOFTWARE DEVELOPMENT SYSTEM

The Micro Works Software Development System (SDS80C) is a complete 6809 editor, assembler and monitor package contained in one Color Computer program pack! Vastly superior to RAM-based assemblers/editors, the SDS80C is non-volatile, meaning that if your application program bombs, it can't destroy your editor/assembler. Plus it leaves almost all of 16K or 32K RAM free for *your* program. Since all three programs, editor, assembler and monitor are co-resident, we eliminate tedious program loading when going back and forth from editing to assembly and debugging!

The powerful screen-oriented Editor features finds, changes, moves, copys and much more. All keys have convenient auto repeat (typamatic), and since no line numbers are required, the full width of the screen may be used to generate well commented code.

The Assembler features *all* of the following: complete 6809 instruction set; complete 6800 set supported for cross-assembly; conditional assembly; local labels; assembly to cassette tape or to memory; listing to screen or printer; and mnemonic error codes instead of numbers.

The versatile ABUG monitor is a compact version of CBUG, tailored for debugging programs generated by the Assembler and Editor. It features examine/change of memory or registers, cassette load and save, breakpoints and more.

SDS80C Price: $89.95

CRRCK THOSE ROMS!

SOURCE GENERATOR: This package is a disassembler which runs on the color computer and enables you to generate your own source listing of the BASIC interpreter ROM. Also included is a documentation package which gives useful ROM entry points, complete memory map, I/O hardware details and more. Disassembler features include cross-referencing of variables and labels; output code which can be reassembled; output to an 80-column printer, small printer or screen; and a data table area specification which defaults to the table boundaries in the interpreter ROM. A 16K system is required for the use of this cassette.

80C Disassembler Price: $49.95

LEARN 6809!

6809 Assembly Language Programming, by Lance Leventhal, contains the most comprehensive reference material available for programming your Color Computer.

Price: $16.95

PARALLEL D!

USE A PARALLEL PRINTER with your Color Computer! Adaptor box plugs into the serial port and allows use of Centronics/Radio Shack compatible printers with parallel interface. Assembled and tested.

PI80C Price: $69.95

CBUG IS HERE!

MONITOR TAPE: A cassette tape which allows you to:

- Examine or change memory using a formatted hex display
- Save areas of memory to cassette in binary (a "CSAVEM")
- Download/upload data or programs to a host system
- Move the video display page throughout RAM
- Send or receive RS-232 at up to 9600 baud
- Investigate and activate features of your computer, such as hi-res graphics or machine-language music
- Use your color computer as an intelligent peripheral for another computer, a color display or a 6809 program development tool

The monitor has 19 commands in all, and is relocatable and re-entrant.

CBUG Tape Price: $29.95

MONITOR ROM: The same program as above, supplied in 2716 EPROM. This allows you to use the entire RAM space. And you don't need to re-load the monitor each time you use it. The EPROM plugs into the Extended Basic ROM Socket or a modified ROMPACK.

CBUG ROM Price: $39.95

32K RAM!

MEMORY UPGRADE KITS: Consisting of 4116 200ns. integrated circuits, with instructions for installation. **4K-16K Kit Price: $39.95. 16K-32K Kit (requires soldering experience) Price: $39.95**

THE MICRO WORKS

MasterCharge/Visa Accepted

California residents add 6% tax.

P.O. BOX 1110, DEL MAR, CA 92014 (714) 942-2400

```
400 PRINT"CHART DATE IS EARLIER
THAN BIRTH";:PRINT"DATE. PLEASE
TRY AGAIN.":FORI=1TO600:NEXTI:GO
TO280
410 FORK=1TO100:NEXTK
420 GOSUB790
430 N=JC-JB
440 V=23:GOSUB880:GOSUB900
450 V=28:GOSUB880:GOSUB900
460 V=33:GOSUB880:GOSUB900
470 GOSUB1070
480 GOSUB1540
490 Z1=Z1+1:PRINT#-2,"* "+C$;TAB
(15)L$;TAB(75)DW$+" *"
500 DW1=DW1+1
510 IF DW1>7 THEN DW1=1
520 JC=JC+1:L=L+1:IF L<E1 THEN 4
30
530 IF Z1>=E1 THEN 1220
540 L=0:GOTO420
550 S4=S4+1:PRINT
560 INPUT"MONTH (1 TO 12)";M
570 IF S4<=1 THEN L1=M
580 M=INT(M):IF M<1 OR M>12 THEN
 560
590 INPUT"DAY (1 TO 31)";D
600 IF S4<=1 THEN L2=D
610 IF ZZ=1 THEN 620 ELSE 640
620 CLS:PRINT@96,"PLEASE ENTER T
HE DAY OF THE WEEK";:PRINT"1=MON
DAY":PRINT"2=TUESDAY":PRINT"3=WE
DNESDAY":PRINT"4=THURSDAY":PRINT
"5=FRIDAY":PRINT"6=SATURDAY":PRI
NT"7=SUNDAY"
630 INPUTDW1
640 D=INT(D):IF D<1 OR D>31  THE
N 590
650 INPUT"YEAR (ENTER ONLY LAST
TWO DIGITS OF THE YEAR)";Y$
654 IF LEN(Y$)>=3 THEN GOTO 650
ELSE GOTO 660
660 Y=VAL(Y$):IF S4<=1 THEN L3=Y
670 IF S4<=1 THEN D4=M+D+Y
680 Y=INT(Y):IF Y<0 THEN 650
690 IF Y>99 THEN 710
700 Y=Y+1900:PRINTY;"ASSUMED.":F
ORI=1TO500:NEXTI:CLS
710 RETURN
720 W=FIX((M-14)/12)
730 JD=INT(1461*(Y+4800+W)/4)
740 B=FIX(367*(M-2-W*12)/12)
750 JD=JD+B
760 B=INT(INT(3*(Y+4900+W)/100)/
4)
770 JD=JD+D-32075-B
780 RETURN
790 IFB$="YES"AND Z>=1 THEN RETU
RN ELSE 800
800 Z=Z+1:CLS
810 PRINT#-2,TAB(36)"BIORHYTHM"
820 PRINT#-2,TAB(39)"FOR"
830 X1=(40-(INT(LEN(N$)/2)))
840 PRINT#-2,TAB(X1)N$;CHR$(10);
CHR$(10):PRINT#-2,"THIS CHART IS
 FOR"E1"DAYS.";TAB(55)"BIRTHDATE
 IS"L1;"/";L2;"/";L3;CHR$(10)
850 PRINT#-2,"  DATE  ";TAB(17);
860 PRINT#-2,"L O W";TAB(40)"O";
TAB(55)"H I G H";TAB(73)"  DAY
"
870 PRINT#-2,STRING$(80,95):RETU
RN
880 W=INT(N/V):R=N-(W*V)
890 RETURN
900 IF V<>23 THEN 950
910 L$=CHR$(32):FOR K=1 TO 5:L$=
L$+L$:NEXT
920 L$=L$+LEFT$(L$,19)
930 L$=LEFT$(L$,T)+CHR$(48)+RIGH
T$(L$,T)
940 IF V=23 THEN C$="P"
950 IF V=28 THEN C$="E"
960 IF V=33 THEN C$="I"
970 W=R/V:W=W*2*P
980 W=T*SIN(W):W=W+T+1.5
990 W=INT(W):A$=MID$(L$,W,1)
1000 IF A$="P" OR A$="E" OR A$="
*" THEN C$="*"
1010 IF W=1 THEN 1050
1020 IF W=T+T+1 THEN 1060
1030 L$=LEFT$(L$,W-1)+C$+RIGHT$(
L$,T+T+1-W)
1040 RETURN
1050 L$=C$+RIGHT$(L$,T+T):RETURN
1060 L$=LEFT$(L$,T+T)+C$:RETURN
1070 W=JC+68569:R=INT(4*W/146097
)
1080 W=W-INT((146097*R+3)/4)
1090 Y=INT(4000*(W+1)/1461001)
1100 W=W-INT(1461*Y/4)+31
1110 M=INT(80*W/2447)
1120 D=W-INT(2447*M/80)
1130 W=INT(M/11):M=M+2-12*W
1140 Y=100*(R-49)+Y+W
1150 A$=STR$(M):W=LEN(A$)-1
1160 C$=MID$(A$,2,W)+"/"
1170 A$=STR$(D):W=LEN(A$)-1
1180 C$=C$+MID$(A$,2,W)+"/"
1190 A$=STR$(Y):W=LEN(A$)-1
1200 C$=C$+MID$(A$,W,2)
1210 RETURN
1220 PRINTTAB(11)"END OF RUN":PR
INT#-2,STRING$(80,95):PRINT#-2,T
AB(36)"END OF RUN"
1230 PRINT#-2,CHR$(31);TAB(17)"T
HANK YOU";CHR$(10);CHR$(30)
1240 CLS:PRINT@224,"ANOTHER RUN
(Y/N)?"
1250 AR$=INKEY$:IFAR$=""THEN1250
1260 IFAR$<>"Y"THEN1270ELSE50
1270 CLS:PRINT@224,"THANKS ANYWA
Y, BYE FOR NOW.":FORI=1TO1000:NE
XTI:CLS:END
```

(Continued on Page 17

STATS (Continued from pg. 8)

```
213 PRINT "ENTER PLAYER'S NUMBER
":PRINTTAB(4)"(OR '99' FOR BOX)"
:PRINT
214 ZA$=INKEY$:IFZA$="" THEN 214
215 ZB$=INKEY$:IFZB$="" THEN 215
216 ZC$=ZA$+ZB$
217 PN=VAL(ZC$)
220 IF PN=HN(1) THEN 300
221 IF PN=HN(2) THEN 311
222 IF PN=HN(3) THEN 312
223 IF PN=HN(4) THEN 313
224 IF PN=HN(5) THEN 314
225 IF PN=HN(6) THEN 315
226 IF PN=HN(7) THEN 316
227 IF PN=HN(8) THEN 317
228 IF PN=HN(9) THEN 318
229 IF PN=HN(10) THEN 319
230 IF PN=HN(11) THEN 320
231 IF PN=HN(12) THEN 321
240 IF PN=99 THEN 4000
250 STOP
300 PH=1:PRINT"PLAYER: "HN$(1):G
OTO1000
311 PH=2:PRINT "PLAYER: "HN$(2):
GOTO1000
312 PH=3:PRINT "PLAYER: "HN$(3):
GOTO1000
313 PH=4:PRINT "PLAYER: "HN$(4):
GOTO1000
314 PH=5:PRINT "PLAYER: "HN$(5):
GOTO1000
315 PH=6:PRINT "PLAYER: "HN$(6):
GOTO1000
316 PH=7:PRINT"PLAYER: "HN$(7):G
OTO1000
317 PH=8:PRINT"PLAYER: "HN$(8):G
OTO1000
318 PH=9:PRINT"PLAYER: "HN$(9):G
OTO1000
319 PH=10:PRINT"PLAYER: "HN$(10)
:GOTO1000
320 PH=11:PRINT"PLAYER: "HN$(11)
:GOTO1000
321 PH=12:PRINT"PLAYER: "HN$(12)
:GOTO1000
400 PRINTV$" STATISTIC"
410 PRINT"  ENTER PLAYER'S NUMBE
R":PRINTTAB(4)"(OR '99' FOR BOX)
":PRINT
420 PRINT
430 ZA$=INKEY$:IF ZA$=""THEN430
433 ZB$=INKEY$:IF ZB$=""THEN433
435 ZC$=ZA$+ZB$
440 PN=VAL(ZC$)
500 IF PN=VN(1) THEN 600
501 IF PN=VN(2) THEN 601
502 IF PN=VN(3) THEN 603
503 IF PN=VN(4) THEN 604
504 IF PN=VN(5) THEN 605
505 IF PN=VN(6) THEN 606
506 IF PN=VN(7) THEN 607
507 IF PN=VN(8) THEN 608
508 IF PN=VN(9) THEN 609
509 IF PN=VN(10) THEN 610
510 IF PN=VN(11) THEN 611
511 IF PN=VN(12) THEN 612
512 IF PN=99 THEN 5000
600 PV=1:PRINT"PLAYER: "VN$(1):G
OTO1900
601 PV=2:PRINT"PLAYER: "VN$(2):G
OTO1900
603 PV=3:PRINT"PLAYER: "VN$(3):G
OTO1900
604 PV=4:PRINT"PLAYER: "VN$(4):G
OTO1900
605 PV=5:PRINT"PLAYER: "VN$(5):G
OTO1900
606 PV=6:PRINT"PLAYER: "VN$(6):G
OTO1900
607 PV=7:PRINT"PLAYER: "VN$(7):G
OTO1900
608 PV=8:PRINT"PLAYER: "VN$(8):G
OTO1900
609 PV=9:PRINT"PLAYER: "VN$(9):G
OTO1900
610 PV=10:PRINT"PLAYER: "VN$(10)
:GOTO1900
611 PV=11:PRINT"PLAYER: "VN$(11)
:GOTO1900
612 PV=12:PRINT"PLAYER: "VN$(12)
:GOTO1900
1000 PRINT:PRINT:PRINTTAB(12)"<E
NTER>"
1001 PRINTTAB(3)"(G)OAL";TAB(17)
"(N)O GOAL"
1002 PRINTTAB(3)"(F)REE THROW";T
AB(17)"(L)INE MISS"
1003 PRINTTAB(10)"(E)NTERED GAME
"
1010 HC$=INKEY$:IF HC$="" THEN 1
010 ELSE PRINT HC$
1015 CLS
1020 IF HC$="G" THEN HG(PH)=HG(P
H)+1:HS(PH)=HS(PH)+1:PA=PA+1:PM=
PM+1
1030 IF HC$="N" THEN HS(PH)=HS(P
H)+1:PA=PA+1
1040 IF HC$="F" THEN HF(PH)=HF(P
H)+1:HL(PH)=HL(PH)+1:FF=FF+1:LT=
LT+1
1045 IF HC$="L" THEN HL(PH)=HL(P
H)+1:LT=LT+1
1047 IF HC$="E" THEN MP(PH)=MP(P
H)+1
1050 GOTO210
1900 PRINT:PRINT:PRINTTAB(12)"<E
NTER>"
1910 PRINTTAB(3)"(G)OAL";TAB(17)
"(N)O GOAL"
1920 PRINTTAB(3)"(F)REE THROW";T
AB(17)"(L)INE MISS"
1930 PRINTTAB(10)"(E)NTERED GAME
"
1950 VC$=INKEY$:IF VC$="" THEN 1
950 ELSE PRINT VC$
1960 CLS
```

(Continued on Page 17)

DOWNLOAD (Continued from pg. 9)

I am sure this combination will open all kinds of different opportunities for you. It has for me.

The listings:

```
1 REM **VIDFIX BY JORGE MIR**
2 REM **      (c) 1981
10 A=30208
20 POKE 2103,255
30 POKE 2112,53
40 FOR X=1728 TO 3839
50 POKE A,PEEK(X)
55 PRINT CHR$(PEEK(X));
60 A=A+1:NEXT X
70 PRINT:PRINT"READY RECORDER"
80 IF INKEY$=""THEN 80
90 FOR X=1 TO 5
100 CSAVEM"VIDEOTEX",30208,
       32319,30208
110 MOTOR ON
120 FOR Z=1 TO 500:NEXT Z
130 MOTOR OFF
140 END


7  **VIDEOAID**
8  **(c) By JORGE MIR, 1981**
10 CLEAR512:PAGE=6:GOSUB120
15 DEFFNA(L)=INT(L/32)*32
20 A$=INKEY$:IFINKEY$=""THEN20
30 IFA$=" "THENPAGE=PAGE+1
40 IFA$="B"THENPAGE=PAGE-1
50 IFA$="D"THEN340
60 IFA$="S"THEN290
70 IFA$="P"THEN240
80 IFA$="L"THEN300
90 IFA$="C"THENGOSUB310
100 GOSUB110:GOTO20
110 IFPAGE>62THENPAGE=0:GOTO20
120 FOR F=0TO1:FOR E=0TO1:FOR D=
0TO1:FORC=0TO1:FOR B=0TO1:FOR A=
0TO1
130 IFPAGE=A*32+B*16+C*8+D*4+E*2
+F THEN150
140 NEXTA,B,C,D,E,F
150 FOR X=65478 TO 65488 STEP2
160 POKEX,200:NEXT X
170 IFF=1THENPOKE65479,100
180 IFE=1THENPOKE65481,100
190 IFD=1THENPOKE65483,100
200 IFC=1THENPOKE65485,100
210 IFB=1THENPOKE65487,100
220 IFA=1THENPOKE65489,100
230 P=PAGE*512:RETURN
240 L=0:A$="":FOR X=PTOP+511
250 A=PEEK(X):IFA=>96ANDA<=127TH
ENA=A-64
260 A$=A$+CHR$(A):L=L+1:IFL<>32T
HEN280
270 PRINT#-2,A$:L=0:A$=""
280 NEXTX:GOTO20
290 CSAVEM"P"+MID$(STR$(PAGE),2)
,P,P+511,P:GOTO20
300 GOSUB530:CLOADMSR$:PAGE=(PEE
K(&H01E7)*256+PEEK(&H01E8))/512:
GOSUB120:GOTO20
310 CLS:PRINT"YOU ARE VIEWING PA
GE#"PAGE
320 PRINT"LOCATED AT"P"("HEX$(P)
")"
330 PRINT:INPUT"ENTER NEW PAGE #
";PAGE:RETURN
340 A1=PEEK(P):POKEP,207:A3=P
350 I$=INKEY$:IFI$=""THEN350
360 I=ASC(I$)
370 IFI$="B"THENBE=P ELSE IF I$=
"E" THENPOKEP+1,255 ELSE IF I$="
L" THEN POKEP+1,13 ELSE IF I$="C
" THEN POKEP+1,96 ELSE IF I$="Q"
 THEN GOSUB600 ELSE IF I=9 THEN
P=P+1
380 IFI=93THENP=INT(P/32)*32+31
ELSE IF I=21 THEN P=INT(P/32)*32
 ELSE IF I=94 THENP=P-32 ELSE IF
 I=10 THEN P=P+32 ELSE IF I=8 TH
EN P=P-1:ELSE IF I=9 THENP=P+1
390 POKE A3,A1:IFI$="E"THEN430
400 IFP>PAGE*512+511THENPAGE=PAG
E+1:P1=P-FNA(P):GOSUB110:P=P+P1
410 IFP<PAGE*512THENPAGE*PAGE-1:
GOSUB110
420 GOTO340
430 CLS:PRINT"GET RECORDER READY
":INPUT"PROGRAM NAME";I$
440 OPEN"O",-1,I$
450 V=PEEK(BE):IFV=255 THEN480
460 IFV=>96 AND V<=127 THEN V=V-
64
470 IFV=13THEN480 ELSE IF V=S TH
EN 490 ELSE L$=L$+CHR$(V):GOTO49
0
480 PRINTL$:PRINT#-1,L$:IFV=255T
HEN510ELSEL$=""
490 IFV=32 THEN S=32 ELSE S=96
500 BE=BE+1:GOTO450
510 CLOSE-1:CLS:PRINT"DOWNLOAD H
AS BEEN COMPLETED"
520 END
530 INPUT "PAGE";PG:IFPG=0THENSR
$=""ELSESR$="P"+MID$(STR$(PG),2)
535 RETURN
600 POKEP,175
610 I$=INKEY$:IFI$=""THEN610
620 A1=ASC(I$):IFA1=>32ANDA1<=63
THENA1=A1+64:P=P+1:RETURN
630 RETURN
640 POKEA3,A1:P=INT(P/32)*32+63:
FOR P=P TO P-32 STEP-1:IFPEEK(P)
=96THENNEXT
641 RETURN
```

STATS (Continued from pg. 15)

```
1970 IF VC$="G" THEN VG(PV)=VG(P
V)+1:VS(PV)=VS(PV)+1:JP=JP+1:JM=
JM+1
1975 IF VC$="N" THEN VS(PV)=VS(P
V)+1:JP=JP+1
1980 IF VC$="F" THEN VF(PV)=VF(P
V)+1:VL(PV)=VL(PV)+1:KT=KT+1:KF=
KF+1
1985 IF VC$="L" THEN VL(PV)=VL(P
V)+1:KT=KT+1
1990 IF VC$="E" THEN VE(PV)=VE(P
V)+1
1995 GOTO210
1997 STOP
2020 GOTO210
4000 CLS:PRINTH$ " BOX SCORE"
4010 PRINTSTRING$(32,"-");
4015 PRINTBZ$;
4020 FOR X=1 TO SS
4025 HT(X)=(HG(X)*2)+HF(X)
4030 IF MP(X)>0 THEN PRINTUSINGB
S$;HN$(X),HG(X),HS(X),HF(X),HL(X
),HT(X)
4035 GH=GH+HG(X):SH=SH+HS(X):FH=
FH+HF(X):LH=LH+HL(X):TH=TH+HT(X)
4040 NEXT X
4045 PRINTUSINGBT$;GH,SH,FH,LH,T
H
4047 IF SH>0 THEN GP=(GH/SH)*100
4048 IF LH>0 THEN FP=(FH/LH)*100
4049 PRINTUSINGPC$;GP,FP
4050 PRINT@480," ";:INPUT "CONTI
NUE";CN$
4060 CLS:GH=0:SH=0:FH=0:LH=0:TH=
0:GOTO210
5000 CLS:PRINTV$" BOX SCORE"
5010 PRINTSTRING$(32,"-");
5015 PRINTBZ$;
5020 FOR X=1 TO VV
5025 VT(X)=(VG(X)*2)+VF(X)
5030 IF VE(X)>0 THEN PRINTUSINGB
S$;VN$(X),VG(X),VS(X),VF(X),VL(X
),VT(X)
5035 GV=GV+VG(X):SV=SV+VS(X):FV=
FV+VF(X):LV=LV+VL(X):TV=TV+VT(X)
5040 NEXT X
5045 PRINTUSINGBT$;GV,SV,FV,LV,T
V
5047 IF SV>0 THEN BP=(GV/SV)*100
5048 IF LV>0 THEN CP=(FV/LV)*100
5049 PRINTUSINGPC$;BP,CP
5050 PRINT@480," ";:INPUT "CONTI
NUE";CN$
5060 CLS:GV=0:SV=0:FV=0:LV=0:TV=
0:GOTO210
```

BIO (Continued from pg. 14)

```
1520 FOR I=1 TO 5:PRINT#-2,CHR$(
10):NEXT I:RETURN
1530 GOTO10
1540 ON DW1 GOTO 1550,1560,1570,
1580,1590,1600,1610
1550 DW$="MON":RETURN
1560 DW$="TUE":RETURN
1570 DW$="WED":RETURN
1580 DW$="THU":RETURN
1590 DW$="FRI":RETURN
1600 DW$="SAT":RETURN
1610 DW$="SUN":RETURN
1650 CLS:PRINT:PRINT:PRINT:PRINT
"NAME : ";N$
1660 PRINT:PRINT"CHART IS FOR ";
E1;" DAYS."
1665 PRINT:PRINT:INPUT"IS THIS I
NFORMATION CORRECT      (Y/N)";X$
1666 IF LEFT$(X$,1)<>"Y" THEN GO
TO 260
1670 RETURN
1700 CLS:PRINT:PRINT:PRINT"MONTH
 :";M
1710 PRINT:PRINT"DAY :";D
1720 PRINT:PRINT"YEAR :";Y
1730 PRINT:INPUT"IS THIS INFORMA
TION CORRECT      (Y/N) :";X$
1735 IF LEFT$(X$,1)<>"Y" THEN GO
TO 300 ELSE RETURN
1750 CLS:PRINT:PRINT:PRINT"MONTH
 :";M
1760 PRINT:PRINT"DAY :";D
1770 PRINT:PRINT"YEAR :";Y
1780 PRINT:INPUT"IS THIS INFORMA
TION CORRECT      (Y/N) :";X$
1790 IF LEFT$(X$,1)<>"Y" THEN GO
TO 340 ELSE RETURN
```